
PL/SQL 4. gyakorlat
PL/SQL függvények használata, függvény csomagok

I. Dátumkezelés, kétdimenziós tömb

Készítsünk tárolt függvényt, amely bemenő paraméterként egy dátumot kap és megmondja, az 
hányadik napja az évnek (szökőévekben a február 29 napos).

create or replace function day_of_year (d IN date) return number is
type month_at is varray(12) of integer;
type year_at is varray(2) of month_at;
days_inmonth year_at := year_at(

month_at(31,28,31,30,31,30,31,31,30,31,30,31),
month_at(31,29,31,30,31,30,31,31,30,31,30,31));

yeartype number(1);
year number(4) := to_number(to_char(d, 'yyyy'));
month number(2) := to_number(to_char(d, 'mm'));
day number(2) := to_number(to_char(d, 'dd'));

begin
-- szökőév minden negyedik, de a századik nem, viszont a négyszázadik igen
if mod(year,4)=0 and mod(year,100)!=0 or mod(year,400)=0 then yeartype:=2;

 else yeartype:=1;
end if;
for i in 1..month-1 loop

day := day + days_inmonth(yeartype)(i);
end loop;
return day;

end;

Hívása:
select day_of_year(sysdate) from dual;
select day_of_year(to_date('2015.03.15.', 'yyyy.mm.dd.')) from dual;

Készítsük el a feladat inverzét. Egy adott év valahanyadik napjáról mondjuk meg, hogy melyik
hónap  melyik  nap.  Például  2016-ban  (szőkőév)  a  60.  nap  február  29.,  de  2017-ben  a  60.  nap
március  1.  Használjuk  az  előző  feladatban  definiált  kétdimenziós  tömböt!  Tárolt  eljárással  és
függvénnyel is oldjuk meg a feladatot! Ügyeljenek a paraméterek típusára (IN, OUT, IN OUT)!

create or replace procedure date_from_day (year IN number, d IN OUT number, month OUT 
number, day OUT number) is

type month_at is varray(12) of integer;
type year_at is varray(2) of month_at;
days_inmonth year_at := year_at(

month_at(31,28,31,30,31,30,31,31,30,31,30,31),
month_at(31,29,31,30,31,30,31,31,30,31,30,31));

yeartype number(1);
            i number(2) := 1;
begin

-- szökőév minden negyedik, de a századik nem, viszont a négyszázadik igen
if mod(year,4)=0 and mod(year,100)!=0 or mod(year,400)=0 then yeartype:=2;

 else yeartype:=1;
end if;



while d > days_inmonth(yeartype)(i) loop
d := d - days_inmonth(yeartype)(i);
i := i + 1;

end loop;
month := i;
day := d;

end;

Hívása:
declare

y number(4) := 2017;
m number(2);
d number(2);
day_of_year number := 60;

begin
date_from_day(y, day_of_year, m, d);
dbms_output.put_line(y || '.' || m || '.' || d || '.');

end;

Ha a feladatot függvénnyel oldjuk meg, egy dátum lehet a visszatérési érték:
create or replace function date_from_day (year IN number, d IN OUT number) return date is 

pl/sql blokk

A visszatérési érték (m_date) előállítása a pl/sql blokkban:
m_date := to_date( to_char(year) || '.' || to_char(month) || '.' || to_char(day) , 'yyyy.mm.dd' );

A függvény hívása, ha van OUT vagy IN OUT típusú paramétere akkor csak blokkból történhet.
Egyébként parancssori SELECT utasításban is meghívható.

II. Sztringkezelés, package

Készítsünk  tárolt  eljárást,  amely  az  autótáblában  eltárolt  rendszámokat  ellenőrzi  formai
szempontból (helyes formátum: három alfabetikus kar. után 3 szám), és kiírja a hibás rendszámokat.
A formátum ellenőrzéshez használjon függvényeket, amelyeket csomagban definiál.

create or replace package format_check is
function is_number(p_string IN varchar2) return int;
function is_alpha(p_string IN varchar2) return int;
end;

create or replace package body format_check is
function is_number (p_string IN varchar2) return int is

   n number;
begin

if length(trim(p_string)) = 3 then
        n := TO_NUMBER(p_string);
   return 1;

else
   return 0;

end if;



exception  --ha a to_number nem tudja elvégezni a konverziót
when value_error then return 0;

end is_number;

function is_alpha (p_string IN varchar2) return int is
n number;

begin
if length(trim(p_string)) = 3 then
      n := LENGTH(TRIM(TRANSLATE(p_string, 

                              'abcdefghijklmnopqrstuvwxyzABCDEFGHIJKLMNOPQRSTUVWXYZ', '  '))) ;
      if (n>0) then return 0;
      else return 1;

              end if;
                        else return 0;
                        end if;

end is_alpha;
end;

--A  TRANSLATE(str1, mit, mivel) fv az első argumentumban megadott sztringben helyettesíti a
--második  argumentumban  megadott  karaktereket  (mit)  a  harmadik  argumentumban  megadott
--karakterekkel  (mivel).  A fenti  példában  üres  karakterrel  (space)  helyettesíti  az  alfanumerikus
--karaktereket.  Majd  levágjuk  az  üres  karaktereket  (  TRIM()  )  és  a  megmaradt  karakterek
--megszámoljuk (karaktersorozat hossza: LENGTH() ).

create or replace procedure rsz_ellenorzo is
rendszam auto.rsz%type;
rsz_alpha char(3);
rsz_num char(3);
db number(2) := 0;
cursor c1 is select rsz from auto;

begin
    dbms_output.put_line('Hibás rendszámok: ');
    open c1;
    loop

fetch c1 into rendszam;
exit when c1%notfound;
select SUBSTR(rendszam,1,3) into rsz_alpha from dual;
select SUBSTR(rendszam,4,3) into rsz_num from dual;
if (format_check.is_alpha(rsz_alpha)=0 or format_check.is_number(rsz_num)=0) then 

db:=db+1;
dbms_output.put_line(rendszam);

        end if;
    end loop;
    close c1;
    dbms_output.put_line('Összesen: ' || db || ' db');
end;

III. Készítsünk ütemezett job-ot. A rendszám ellenőrző tárolt eljárás fusson le a mai naptól 
számítva 30 napon át 12 óránként.
Job: időponthoz kötött (ütemezett) tevékenységvégrehajtás
Kezelése: gyári fvcsomagon keresztül



BEGIN
DBMS_SCHEDULER.CREATE_JOB (
   job_name         =>  'rsz_ell',
   job_type           =>  'STORED_PROCEDURE',
   job_action        =>  'TEST.RSZ_ELLENORZO',   --a tárolt eljárás neve
   start_date         =>  SYSTIMESTAMP,
   repeat_interval =>  'FREQ=HOURLY; INTERVAL=12',
   end_date           =>  SYSTIMESTAMP + INTERVAL '30' day,
   comments         =>  'My new job');
END;

Előredefiniált gyakoriságok: YEARLY, MONTHLY, WEEKLY, HOURLY, MINUTELY, 
SECONDLY

Egyéb függvények:
Job jellemző értékének beállítása: 
    DBMS_SCHEDULER.SET_JOB_ARGUMENT_VALUE (job_name, argument_name, value);
Job jellemző törlése:
    DBMS_SCHEDULER.RESET_JOB_ARGUMENT_VALUE (job_name, argument_name);
Job törlése:
     DBMS_SCHEDULER.DROP_JOB(job_name, TRUE/FALSE);
     TRUE: éppen futó jobot is töröl
     FALSE: default, éppen futó jobot nem töröl (hibaüzenettel tér vissza)

IV. Merge parancs

Hozzunk létre egy új autó táblát, amely az autó táblával megegyező szerkezetű azzal a kivétellel,
hogy az évjárat  mező helyett  az autó kora legyen megadva.  Készítsünk tárolt  eljárást,  amely a
MERGE parancs  felhasználásával  az  autó  tábla  évjárat  mezőértékei  alapján  kitölti  a  kor  mező
értékeit.

--Új autó tábla létrehozása az adatok másolásával. Az eljárás WHEN MATCHED feltétele teljesül.
--Figyelem! Az integritási feltételek nem másolódnak!
create table ujauto as select rsz, tipus, szin, ar from auto;
alter table ujauto modify(rsz primary key);
alter table ujauto add(kor number(2) check (kor>0));

--Új autó tábla létrehozása üresen. Az eljárás WHEN NOT MATCHED feltétele teljesül.
create table ujauto( rsz char(6) primary key, tipus char(10) not null, szin char(10) default 'piros', ar
number(8), kor number(2) check (kor>0) );

create or replace procedure korszamito is
begin
   merge into ujauto using auto on (ujauto.rsz = auto.rsz)
   when matched then
      update set ujauto.kor = to_number(to_char(sysdate, 'yyyy'))-auto.evjarat
   when not matched then
      insert (rsz, tipus, szin, ar, kor)
      values(auto.rsz,auto.tipus,auto.szin,auto.ar,

      to_number(to_char(sysdate,'yyyy'))-auto.evjarat);
end;



Sztringkezelő és karakterkonverziós függvények:

ASCII(karakter) Visszaadja a karakter ASCII kódját.

CHR(ASCII kód) Visszaadja a megadott ASCII kódhoz tartozó karaktert.

COMPOSE(string) Például ékezetes karakterek előállításához:
COMPOSE('o' || unistr('\0308') ) → Result: ö
COMPOSE('e' || unistr('\0301') ) → Result: é

CONCAT(str1, str2) Két sztring összefűzése (uaz a szerepe mint a || operátornak).

INITCAP(string) A megadott sztring minden szavát átalakítja nagy kezdőbetűsre (a többi 
betű kicsi lesz).

INSTR(str, substr, 
start, hanyadik)

Az első sztringben keresi a második sztring előfordulását. Ha meg van adva
a start pozíció, akkor innentől keres (default 1). Opcionálisan megadható h.
hanyadik előfordulást keresse (default 1).

LENGTH(string) Sztring hossza.

LOWER / UPPER 
(str)

A megadott sztring kis- ill. nagybetűssé konvertálása.

LPAD / RPAD (str, 
length, töltőstr)

Kiegészíti a megadott sztringet (balról ill. jobbról) az adott méretűre. Ha a 
feltöltő sztring nincs megadva, akkor space-ekkel egészíti ki.

TRIM / LTRIM / 
RTRIM (str, substr)

A megadott sztringből levágja a második paraméterben adott sztringet. Ha 
ez utóbbi üres, akkor a space-eket vágja le.

SUBSTR(str, start, 
length)

A sztringből a megadott pozíciótól kezdve adott hosszúságú részsztringet 
ad vissza. Ha ez utóbbi paraméter nincs megadva, akkor az adott pozíciótól
hátralévő részsztringet adja vissza.

REPLACE(str, mit, 
mivel)

A megadott sztringben lecseréli a második paraméterként megadott rész-
sztringet a harmadik paraméterben adott sztringre. Ha ez utóbbi nincs meg-
adva, akkor törli a sztringből a második paraméterben adott részsztringet.
Pl.: REPLACE('123tech123', '123'); → Result:'tech'

TRANSLATE(str, 
mit, mivel)

A megadott sztringben lecseréli a második paraméterként megadott 
karaktereket a harmadik paraméterben adott karakterekre EGYESÉVEL.
Pl: TRANSLATE('222tech', '2ec', '3it') → Result: '333tith'


