
PL/SQL 2. gyakorlat
SQL utasítások megadása PL/SQL blokkban

I. Ismétlés: SQL DDL, DML utasítások

DDL utasítások: create, drop, alter

Hozzunk létre egy autó táblát.
create table auto ( 

rsz char(6) primary key, 
tipus char(10) not null, 
szin char(10) default 'feher', 
evjarat number(4), 
ar number(8) check(ar>0) 

);
--describe auto;
create table ujauto as select … ; --táblasémát+adatokat átmásol, de integr. feltételeket nem

Módosítsuk a szerkezetét.
alter table add(kor number(2)); --új mező hozzáadása
alter table modify(kor check(kor>0)); --mező típusának/integr.feltételének módosítása
alter table drop(kor); --mező törlése

Tábla törlése:
drop table auto;

DML utasítások: insert, update, delete

Vigyünk fel új rekordot:
insert into auto values ('kda234','opel','piros',2014,1650000);
insert into auto values ('lpd012','mazda',null,2016,2800000); --üres szín
insert into auto (rsz, tipus, evjarat, ar) values ('ksw345','ford', 2009, 15000000); --default szín

Adatmódosítás:
update auto set ar=ar*0.8 where kor>10;

Az átlagnál fiatalabb autók árának növelése.
create view vkor as select rsz, to_char(sysdate, 'yyyy')-evjarat kor from auto;
update auto set ar=ar*1.1 where rsz in (select rsz from vkor where kor < (select avg(kor) 
from vkor));

Adattörlés:
delete from auto where szin='piros';

Az átlagnál idősebb autók törlése.
delete from auto where rsz in (select rsz from vkor where kor > (select avg(kor) from vkor));



II. DML utasítások megadása PL/SQL blokkban

1. Vigyünk fel egyszerre több rekordot az autó táblába.

begin
insert into auto values ('jad123','skoda','piros',11,800000);
insert into auto values ('jzy456','opel','fekete',10,800000);

end;

2. Egy memóriaváltozóban megadott kornál idősebb autók árát csökkentsük 10%-al.

declare
x number(2) := 8;  

begin
update auto set ar=ar*0.9 where rsz in (select rsz from vkor where kor > x);
--a felhasználó adja meg x értékét

           update auto set ar=ar*0.9 where rsz in (select rsz from vkor where kor > :x);
end;

3. Töröljünk egy memóriaváltozóban megadott rendszámú autót.

declare
r auto.rsz%type := 'jzy456';

begin
delete from auto where rsz=r;

end;

Speciális PL/SQL változó típusok:
r auto.rsz%type; --valamely tábla adott mezőjének típusával megegyező
a auto%rowtype; --valamely tábla rekordtípusával megegyező

III. DDL utasítások megadása PL/SQL blokkban (dinamikus parancsvégrehajtás)

4./a Módosítsuk az autó tábla szerkezetét. Vegyünk fel egy sorszám mezőt és adjunk hozzá 
integritási feltételt.

declare
stmt char(100) := 'alter table auto add (sorszam integer)';

begin
execute immediate stmt;
execute immediate 'alter table auto modify (sorszam check(sorszam>0))';

end;

4./b Töröljük az utólag hozzáadott oszlopot.

begin
execute immediate 'alter table auto drop(sorszam)';

end;



IV. DQL utasítások megadása PL/SQL blokkban

Ha a lekérdezés eredménye egy rekord, akkor az eredmény tárolására szolgáló memóriaváltozók 
deklarálása után: 

SELECT projekciós rész INTO memóriaváltozó_lista FROM táblanév;

5. Irassuk ki egy adott rendszámú autó adatait.

declare
a auto%rowtype;
r auto.rsz%type := 'jzy456';

begin
select * into a from auto where rsz=r;
dbms_output.put_line(a.rsz || '  ' || a.tipus || '  ' || a.szin ||'  ' || a.evjarat|| '  ' || a.ar);

end;

Ha a lekérdezés eredménye több rekord, kurzor szerkezetet kell használni. Fajtái: implicit, explicit.

Explicit kurzor létrehozásának, feldolgozásának lépései:
1. Kurzor deklarálása:

CURSOR cnév [ (paraméterlista) ] IS SELECT …;
2. Kurzor megnyitása:

OPEN cnév [ (aktuális paraméterlista) ]; 
3. Eredményrekordok lekérdezése, feldolgozása ciklusban:

LOOP
FETCH cnév INTO memóriaváltozó_lista;
--a ciklusból akkor lépünk ki ha nincs több feldolgozandó rekord
EXIT WHEN cnév%NOTFOUND;

   …
END LOOP;

4. Kurzor lezárása:
CLOSE cnév;

Implicit kurzor: nem kell deklarálni, megnyitni és lezárni; a FOR ciklus utasítás fejlécében 
közvetlenül megadható a lekérdezés.

Ha a lekérdezés feldolgozásakor valamely mező(k) értékét módosítani/törölni szeretnék, a kurzor 
deklarációjában meg kell adni:

CURSOR cnév ... FOR UPDATE OF mezőnévlista; 
Ekkor az eredményrekordok feldolgozásakor, a módosítandó/törlendő rekordra hivatkozáshoz 
felhasználható a kurzor aktuális értéke:

UPDATE / DELETE … WHERE CURRENT OF cnév;

6./a Kérdezzük le a piros színű autók adatait. Explicit kurzorral.

declare
cursor pauto is select * from auto where szin='piros';
a auto%rowtype;

begin
  open pauto;
   loop



    fetch pauto into a;
      exit when pauto%notfound;
      dbms_output.put_line(a.rsz || '  ' || a.tipus || '  ' || a.szin ||'  ' || a.evjarat || '  ' || a.ar);
   end loop;
  close pauto;
end;

6./b Piros autók adatainak lekérdezése. Implicit kurzorral.

begin
    for c in (select * from auto where szin='piros') loop
      dbms_output.put_line(c.rsz || '  ' || c.tipus || '  ' || c.szin ||'  ' || c.evjarat || '  ' || c.ar);
   end loop;
end;

6./c Piros autók adatainak lekérdezése. Az autó színét paraméterként adjuk át a kurzornak.

declare
cursor pauto (sz char) is select * from auto where szin=sz;
a auto%rowtype; 

begin
  open pauto('piros');
   loop
    fetch pauto into a;
      exit when pauto%notfound;
      dbms_output.put_line(a.rsz || '  ' || a.tipus || '  ' || a.szin ||'  ' || a.evjarat || '  ' || a.ar);
   end loop;
  close pauto;
end;

Implicit kurzorral:

declare
    mszin auto.szin%type;
    sz auto.szin%type;
begin
    sz := :mszin || '%' ;  --a változók típusa char(10); a ki nem töltött karaktereket a % fedi le
    for c in (select * from auto where szin like sz) loop
    --VAGY
    --sz := trim(:mszin);
    --for c in (select * from auto where szin = sz) loop
      dbms_output.put_line(c.rsz || '  ' || c.tipus || '  ' || c.szin ||'  ' || c.evjarat || '  ' || c.ar);
   end loop;
end;

Megjegyzés: numerikus adat esetén egyszerűen
for c in (select * from auto where ar=:mar) loop … end loop;

Megjegyzés: a mezőértékek egyezőségének vizsgálata kis- és nagybetű érzékeny.



6./d Piros autók adatainak lekérdezése és árának 10%-os növelése. Az autó színét paraméterként 
adjuk át a kurzornak.

declare
cursor pauto (sz char) is select * from auto where szin=sz for update of ar;
a auto%rowtype; 

begin
  open pauto('piros');
   loop
    fetch pauto into a;
      exit when pauto%notfound;

--módosítás előtt
dbms_output.put_line(a.rsz || '  ' || a.tipus || '  ' || a.szin ||'  ' || a.evjarat|| '  ' || a.ar);

      update auto set ar=ar*1.1 where current of pauto;
      --módosítás után

select * into a from auto where rsz=a.rsz;
                       dbms_output.put_line(a.rsz || '  ' || a.tipus || '  ' || a.szin ||'  ' || a.evjarat|| '  ' || a.ar);
   end loop;
  close pauto;
end;

6./e Piros autók adatainak lekérdezése és árának 10%-os növelése. Implicit kurzorral.

begin
    for c in (select * from auto where szin='piros') loop
            --módosítás előtt
            dbms_output.put_line(c.rsz || '  ' || c.tipus || '  ' || c.szin ||'  ' || c.evjarat || '  ' || c.ar);
      update auto set ar=ar*1.1 where auto.rsz=c.rsz;
      --módosítás után

select * into c from auto where rsz=c.rsz;
dbms_output.put_line(c.rsz || '  ' || c.tipus || '  ' || c.szin ||'  ' || c.evjarat || '  ' || c.ar);

    end loop;
end;

6./f 10 évnél idősebb autók törlése kurzorral.

Explicit kurzor esetén, ha a kurzor feldolgozásakor a WHERE CURRENT OF cursor szelekciós
feltételt  akarjuk  használni,  akkor  rekord  törléséhez  is  a  FOR  UPDATE  OF  kulcsszóval  kell
deklarálni a kurzort.

declare
 mkor number(2) := 10;
   cursor c (kor number) is select * from auto where to_char(sysdate,'yyyy')-evjarat > kor for 

update of rsz;
   a auto%rowtype;
begin
   open c(mkor);
   loop
      fetch c into a;   
      exit when c%notfound;
      delete from auto where current of c;



--ha így törlünk, akkor nem kell a kurzor deklarációban a FOR UPDATE OF
--delete from auto where rsz=a.rsz; 

      dbms_output.put_line('torolve');
    end loop;
            close c;
end;

Implicit kurzorral:

declare
   mkor number(2) := 10;
begin
   for c in (select * from auto where to_char(sysdate,'yyyy')-evjarat > mkor) loop
         delete from auto where rsz=c.rsz;
      dbms_output.put_line('torolve');
    end loop;
end;

Házi feladat:  Piros autók adatainak lekérdezése és árának 10%-os növelése ha 10 évnél fiatalabb;
egyébként töröljük. Explicit és implicit kurzorral is valósítsa meg! 


