
PL/SQL 3. gyakorlat
Hibakezelés, tárolt eljárások/függvények

I. Hibakezelés

Hibakezelő résszel bővített pl/sql blokk:

BEGIN
…
EXCEPTION

WHEN hibakod THEN muvelet;
END;

Leggyakrabban felmerülő gyári hibakódok:
CURSOR_ALREADY_OPEN: már létező kurzort  próbál  újra  megnyitni.  Pl.  ha implicit  kurzor
esetén a for cikluson belül próbljuk megnyitni a kurzort (a for ciklus automatikusan megnyitja).
INVALID_CURSOR:  nem  megengedett  kurzor  művelet  (pl.  olyan  kurzor  bezárása,  ami  nincs
megnyitva).
INVALID_NUMBER: karaktersorozat numerikus értékre történő konverziója nem lehetséges, mert
a karaktersorozat nem számot reprezentál (SQL utasítás esetén).
NO_DATA_FOUND: a SELECT INTO utasítás nem ad vissza rekordot.
TOO_MANY_ROWS: a  SELECT INTO utasítás egynél több rekordot ad vissza.
VALUE_ERROR: aritmetikai vagy konverziós hiba (procedurális utasítások esetén); vagy adattípus
méret megkötés megsértése.
ZERO_DIVIDE: nullával próbáltunk osztani.

Saját hibakód használata:

DECLARE
sajathibakod EXCEPTION;

BEGIN
…
RAISE sajathibakod;
...

EXCEPTION
WHEN sajathibakod THEN muvelet;

END;

Megjegyzés:  a  gyári  hibakódok használhatók nem definiált  hibák jelzésére  is.  Ezeket  nem kell
deklarálni, de jelezni kell a felmerülés helyét (RAISE) és írhatunk saját hibakezelő részt.

--insert into auto values('aaa222', 'skoda', null, 2006, 800000);
declare

sz auto.szin%type;
begin
    select szin into sz from auto where rsz='aaa222'; --létező rekord, ahol a szín üres
   if sz is null then 
         RAISE NO_DATA_FOUND;
     end if;

EXCEPTION
    WHEN NO_DATA_FOUND THEN dbms_output.put_line('A szin nincs megadva');
end;



II. Tárolt eljárás

Adatbázisban tárolt, nevesített PL/SQL blokk. Explicite meg kell hívni.

CREATE OR REPLACE PROCEDURE pnev (paraméternév IN/OUT/IN OUT típus) IS
PL/SQL blokk;

1. Készítsünk tárolt eljárást, amely lekérdezi a piros autók adatait SELECT … INTO paranccsal. Ha
a  lekérdezés  eredménye  egy  rekordnál  több  TOO_MANY_ROWS  hibakezelés,  ha  nincs  a
feltételnek megfelelő rekord NO_DATA_FOUND hibakezelés.

create or replace procedure pauto is
a auto%rowtype;

begin
select * into a from auto where szin='piros';
exception

when too_many_rows then 
dbms_output.put_line('Egynél több rekordot cursorban lehet lekérdezni');

when no_data_found then
dbms_output.put_line('Nincs a feltételnek megfelelő rekord a táblában');

end;

Hívása:

begin
pauto;

end;

Módosítsuk úgy a feladatot, h. a színt input paraméterként kapja meg az eljárás. Figyelem! Az input
(IN) paraméter értéke nem változtatható meg a blokkon belül. A típus nélkül megadott paraméter
alapértelmezés szerint IN típusú.

2. Készítsünk tárolt eljárást, amely csökkenti a 10 évnél idősebb autók árát 20%-al. A kort az eljárás
input paraméterként kapja meg. Az eljárás output paraméterben adja vissza a művelet által érintett
rekordok számát (SQL%ROWCOUNT). Írjon ki hibaüzenetet, ha ez a szám 0.

create or replace procedure oregauto(hatar IN number, rek_szam OUT number) is
zero_record exception;

begin
update auto set ar=ar*0.8 where to_char(sysdate,’yyyy’)-evjarat>hatar;
rek_szam := sql%rowcount;
if sql%notfound then --if rek_szam=0 then

raise zero_record;
            elsif sql%found then --elsif rek_szam>0 then

dbms_output.put_line('Siker!');
            end if;

exception
when zero_record then dbms_output.put_line('Nincs a feltetelnek megfelelo rekord');

end;



Hívása:

declare
akor number(2) := 10;
rszam number(2);

begin
oregauto(akor, rszam);
dbms_output.put_line('A módosított rekordok száma: ' || rszam);

end;

3. Készítsünk tárolt eljárást, amelyben felviszünk egy megadott korú autó rekordot. A kort az eljárás
IN OUT paraméterként  kapja  meg.  A 20-nál  nagyobb  bemenő értéket  hibának tekintjük.  Ilyen
esetben módosítsuk programból a kor értéket 1 és 20 közöttire, és ezt adjuk vissza. Az eljárást hívó
blokkból irassuk ki az adatbázisban eltárolt kor értéket.

create or replace procedure autokor (k IN OUT number) is
l_seed char(100);
invalid_value exception;

begin
if k<1 or k>20 then

raise invalid_value;
else

   insert into auto values('abd123', 'mazda', 'piros', to_char(sysdate,'yyyy')-k, 1200000);
end if;

     exception
when invalid_value then
    l_seed := to_char(SYSTIMESTAMP, 'YYYYDDMMHH24MISSFFFF');
    DBMS_RANDOM.seed(l_seed);
    select dbms_random.value(1,20) into k from dual;  --k értéke módosul
    insert into auto values('abd123', 'mazda', 'piros', to_char(sysdate,'yyyy')-k, 1200000);

end;

Tárolt eljárás hívása:

declare
x number(2) := 25;

begin
autokor(x);    
dbms_output.put_line('Az új autó kora: ' || x || ' év ');

end;

4.  Töltsünk  fel  egy  N elemű tömböt  véletlenszámokkal  és  keressük  meg  a  tömb legkisebb  és
legnagyobb elemét.

create or replace procedure tomb (n in number, mini out integer, maxi out integer) is
--type ttipus is varray(10) of integer;       
--A tömb mérete konstans (fix), nem lehet 'n'
--Megoldás: halmaz típus használata, ami nem fix méretű: TABLE OF integer



        type ttipus is table of integer;
telem ttipus := ttipus();     --konstruktor hívása
l_seed char(100);

begin
l_seed := to_char(SYSTIMESTAMP, 'YYYYDDMMHH24MISSFFFF');
DBMS_RANDOM.seed(l_seed);
for i in 1..n loop    

telem.extend(1); 
select dbms_random.value(-100,100) into telem(i) from dual;
dbms_output.put_line(i || '. elem: ' || telem(i));    --tömbelemek kiírása

end loop;

mini := telem(1);
maxi := telem(1);
for i in 1..n loop

if mini>telem(i) then mini := telem(i); end if;
if maxi<telem(i) then maxi := telem(i); end if;

end loop;
end;

Hívása:

declare
tmin integer;
tmax integer;
meret integer;

begin
tomb(10,tmin,tmax);
dbms_output.put_line('A tömb legkisebb eleme: ' || tmin || ' ; legnagyobb eleme: ' || tmax);

end;

III. Tárolt függvény

Adatbázisban tárolt, nevesített PL/SQL blokk. Explicite meg kell hívni.

CREATE  OR  REPLACE  FUNCTION fnev  (paraméternév  IN/OUT/IN  OUT  típus)  
RETURN típus IS

PL/SQL blokk;

5. Készítsen tárolt függvényt, amely kiszámítja egy paraméterben megadott N szám faktoriálisát.

create or replace function faktorialis (n in number) return number is
fakt number(8) := 1;

begin
for i in 1..n loop    --faktoriális számítás

fakt := fakt * i;
end loop;
return fakt; --fv visszatérési érték

end;



Függvény hívása:

1) select fvnev(par) from dual; --dinamikus sql parancsvégrehajtáshoz nem használható
2) blokkból

declare
x number(2);
y number(8);

begin
x := 5;
y := faktorialis(x);  dbms_output.put_line(x || ' faktorialisa: ' || y);
--VAGY segédváltozó nélkül: dbms_output.put_line(x || ' faktorialisa: ' ||  faktorialis(x));

end;

Tárolt eljárások/függvények lekérdezése:
1)  SELECT object_name  FROM  user_objects  WHERE  object_type  LIKE 'PROCEDURE'  OR
object_type LIKE 'FUNCTION';
2) SELECT text FROM user_source WHERE name='NEV' ORDER BY LINE;

Tárolt eljárások/függvények törlése:
DROP PROCEDURE / FUNCTION nev;


